
		       The Artistic License 2.0

	    Copyright (c) 2000-2006, The Perl Foundation.

     Everyone is permitted to copy and distribute verbatim copies
      of this license document, but changing it is not allowed.

Preamble

This license establishes the terms under which a given free software
Package may be copied, modified, distributed, and/or redistributed.
The intent is that the Copyright Holder maintains some artistic
control over the development of that Package while still keeping the
Package available as open source and free software.

You are always permitted to make arrangements wholly outside of this
license directly with the Copyright Holder of a given Package.  If the
terms of this license do not permit the full use that you propose to
make of the Package, you should contact the Copyright Holder and seek
a different licensing arrangement. 

Definitions

    "Copyright Holder" means the individual(s) or organization(s)
    named in the copyright notice for the entire Package.

    "Contributor" means any party that has contributed code or other
    material to the Package, in accordance with the Copyright Holder's
    procedures.

    "You" and "your" means any person who would like to copy,
    distribute, or modify the Package.

    "Package" means the collection of files distributed by the
    Copyright Holder, and derivatives of that collection and/or of
    those files. A given Package may consist of either the Standard
    Version, or a Modified Version.

    "Distribute" means providing a copy of the Package or making it
    accessible to anyone else, or in the case of a company or
    organization, to others outside of your company or organization.

    "Distributor Fee" means any fee that you charge for Distributing
    this Package or providing support for this Package to another
    party.  It does not mean licensing fees.

    "Standard Version" refers to the Package if it has not been
    modified, or has been modified only in ways explicitly requested
    by the Copyright Holder.

    "Modified Version" means the Package, if it has been changed, and
    such changes were not explicitly requested by the Copyright
    Holder. 

    "Original License" means this Artistic License as Distributed with
    the Standard Version of the Package, in its current version or as
    it may be modified by The Perl Foundation in the future.

    "Source" form means the source code, documentation source, and
    configuration files for the Package.

    "Compiled" form means the compiled bytecode, object code, binary,
    or any other form resulting from mechanical transformation or
    translation of the Source form.


Permission for Use and Modification Without Distribution

(1)  You are permitted to use the Standard Version and create and use
Modified Versions for any purpose without restriction, provided that
you do not Distribute the Modified Version.


Permissions for Redistribution of the Standard Version

(2)  You may Distribute verbatim copies of the Source form of the
Standard Version of this Package in any medium without restriction,
either gratis or for a Distributor Fee, provided that you duplicate
all of the original copyright notices and associated disclaimers.  At
your discretion, such verbatim copies may or may not include a
Compiled form of the Package.

(3)  You may apply any bug fixes, portability changes, and other
modifications made available from the Copyright Holder.  The resulting
Package will still be considered the Standard Version, and as such
will be subject to the Original License.


Distribution of Modified Versions of the Package as Source 

(4)  You may Distribute your Modified Version as Source (either gratis
or for a Distributor Fee, and with or without a Compiled form of the
Modified Version) provided that you clearly document how it differs
from the Standard Version, including, but not limited to, documenting
any non-standard features, executables, or modules, and provided that
you do at least ONE of the following:

    (a)  make the Modified Version available to the Copyright Holder
    of the Standard Version, under the Original License, so that the
    Copyright Holder may include your modifications in the Standard
    Version.

    (b)  ensure that installation of your Modified Version does not
    prevent the user installing or running the Standard Version. In
    addition, the Modified Version must bear a name that is different
    from the name of the Standard Version.

    (c)  allow anyone who receives a copy of the Modified Version to
    make the Source form of the Modified Version available to others
    under
		
	(i)  the Original License or

	(ii)  a license that permits the licensee to freely copy,
	modify and redistribute the Modified Version using the same
	licensing terms that apply to the copy that the licensee
	received, and requires that the Source form of the Modified
	Version, and of any works derived from it, be made freely
	available in that license fees are prohibited but Distributor
	Fees are allowed.


Distribution of Compiled Forms of the Standard Version 
or Modified Versions without the Source

(5)  You may Distribute Compiled forms of the Standard Version without
the Source, provided that you include complete instructions on how to
get the Source of the Standard Version.  Such instructions must be
valid at the time of your distribution.  If these instructions, at any
time while you are carrying out such distribution, become invalid, you
must provide new instructions on demand or cease further distribution.
If you provide valid instructions or cease distribution within thirty
days after you become aware that the instructions are invalid, then
you do not forfeit any of your rights under this license.

(6)  You may Distribute a Modified Version in Compiled form without
the Source, provided that you comply with Section 4 with respect to
the Source of the Modified Version.


Aggregating or Linking the Package 

(7)  You may aggregate the Package (either the Standard Version or
Modified Version) with other packages and Distribute the resulting
aggregation provided that you do not charge a licensing fee for the
Package.  Distributor Fees are permitted, and licensing fees for other
components in the aggregation are permitted. The terms of this license
apply to the use and Distribution of the Standard or Modified Versions
as included in the aggregation.

(8) You are permitted to link Modified and Standard Versions with
other works, to embed the Package in a larger work of your own, or to
build stand-alone binary or bytecode versions of applications that
include the Package, and Distribute the result without restriction,
provided the result does not expose a direct interface to the Package.


Items That are Not Considered Part of a Modified Version 

(9) Works (including, but not limited to, modules and scripts) that
merely extend or make use of the Package, do not, by themselves, cause
the Package to be a Modified Version.  In addition, such works are not
considered parts of the Package itself, and are not subject to the
terms of this license.


General Provisions

(10)  Any use, modification, and distribution of the Standard or
Modified Versions is governed by this Artistic License. By using,
modifying or distributing the Package, you accept this license. Do not
use, modify, or distribute the Package, if you do not accept this
license.

(11)  If your Modified Version has been derived from a Modified
Version made by someone other than you, you are nevertheless required
to ensure that your Modified Version complies with the requirements of
this license.

(12)  This license does not grant you the right to use any trademark,
service mark, tradename, or logo of the Copyright Holder.

(13)  This license includes the non-exclusive, worldwide,
free-of-charge patent license to make, have made, use, offer to sell,
sell, import and otherwise transfer the Package with respect to any
patent claims licensable by the Copyright Holder that are necessarily
infringed by the Package. If you institute patent litigation
(including a cross-claim or counterclaim) against any party alleging
that the Package constitutes direct or contributory patent
infringement, then this Artistic License to you shall terminate on the
date that such litigation is filed.

(14)  Disclaimer of Warranty:
THE PACKAGE IS PROVIDED BY THE COPYRIGHT HOLDER AND CONTRIBUTORS "AS
IS" AND WITHOUT ANY EXPRESS OR IMPLIED WARRANTIES. THE IMPLIED
WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE, OR
NON-INFRINGEMENT ARE DISCLAIMED TO THE EXTENT PERMITTED BY YOUR LOCAL
LAW. UNLESS REQUIRED BY LAW, NO COPYRIGHT HOLDER OR CONTRIBUTOR WILL
BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, OR CONSEQUENTIAL
DAMAGES ARISING IN ANY WAY OUT OF THE USE OF THE PACKAGE, EVEN IF
ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

#! /Users/jj/bin/perl

# a2pdf - converts ASCII text to PDF format, with optional 
#         line/page numbering and Perl syntax highlighting
#
# Copyright (C) 2007 Jon Allen <jj@jonallen.info>
#
# This software is licensed under the terms of the Artistic
# License version 2.0.
#
# For full license details, please read the file 'artistic-2_0.txt' 
# included with this distribution, or see
# http://www.perlfoundation.org/legal/licenses/artistic-2_0.html

use strict;
use warnings;
use App::a2pdf;
use Getopt::Long;
use Pod::Usage;

our $VERSION = '1.13';

binmode *STDOUT;

# Set default options
my %options = (
  header        => 1,          # Include header on all pages
  footer        => 1,          # Include footer on all pages
  line_numbers  => 1,          # Print line numbers
  page_width    => 595,        # A4
  page_height   => 842,        # A4
  left_margin   => 48,         # 0.75"
  right_margin  => 48,         # 0.75"
  top_margin    => 60,         # 
  bottom_margin => 60,         # 
  font_face     => 'Courier',  # Monospaged text
  font_size     => 9,          # Text size = 9 points
  perl_syntax   => 1,          # Perform Perl syntax highlighting
  icon_scale    => 0.5,        # Icon scaling (%age)
  );
  
# Parse command line options
GetOptions(\%options,
  qw/
    font_size|font-size=i
    font_face|font-face=s
    line_spacing|line-spacing=i
    page_size|page-size=s
    page_height|page-height=i
    page_width|page-width=i
    left_margin|left-margin=i
    right_margin|right-margin=i
    top_margin|top-margin=i
    bottom_margin|bottom-margin=i
    header!
    footer|page_numbers|page-numbers!
    line_numbers|line-numbers!
    perl_syntax|perl-syntax|perl!
    title=s
    icon=s
    icon_scale|icon-scale=s
    timestamp
    noformfeed
    /,
  'output_file|o|output-file=s' => sub{open STDOUT,">$_[1]" or die("Cannot open output file $_[1]: $!\n")},
  'margins=i'       => sub{$options{top_margin}    = $_[1];
                           $options{bottom_margin} = $_[1];
                           $options{left_margin}   = $_[1];
                           $options{right_margin}  = $_[1]},
  'version' => sub{print "This is a2pdf, version $VERSION\n"; exit},
  'help'    => sub{pod2usage(-verbose => 1, -noperldoc => 1)},
  'doc'     => sub{pod2usage(-verbose => 2, -noperldoc => 1)}
  ) or die;

$options{title}  = (@ARGV) ? $ARGV[0] : 'STDIN' unless (exists $options{title});
$options{title} .= (' - ' . ((@ARGV) ? scalar localtime($^T - (-M $ARGV[0])*24*60*60) : scalar localtime)) if (exists $options{timestamp});

# Create PDF object
my @text = (<>);
my $pdf  = App::a2pdf->new(%options);
$pdf->line_number_chars(($options{line_numbers}) ? length sprintf("%d",scalar @text) : 0);
  
# Print document
if ($options{perl_syntax}) {
  PERL: {
    eval "use Perl::Tidy";
    if ($@) {
      warn "Cannot perform syntax highlighting, Perl::Tidy not installed\n";
      goto NOPERL;
    }
    Perl::Tidy::perltidy(
      source    => \@text,
      formatter => $pdf
    );
  }
} else {
  NOPERL: {
    foreach my $line (@text) {
      $pdf->print($line);
    }
  }
}

$pdf->output;

exit;


#-----------------------------------------------------------------------

sub _MANIFEST {
  require Perl::Tidy;
}


=head1 NAME

a2pdf - converts ASCII text to PDF format, with optional line/page numbering and Perl syntax highlighting

=head1 SYNOPSIS

 a2pdf [options] [filename]

=head2 Options

B<a2pdf> recognises the following command line options:

=over 4

=item --help

Displays usage message and exits.

=item --doc

Displays full documentation and exits.

=item --version

Prints the version number and exits

=item --output-file | -o

Specifies the filename for the PDF file. If this option is not set,
B<a2pdf> will output to STDOUT.

=item --title

Sets the title to be included in the page header. If unspecified, the
title will default to the name of the file which is being converted,
or to 'STDIN' if B<a2pdf> is processing from standard input.

=item --timestamp

Boolean option - if set, the timestamp of the file to be converted will
be included in the page header. This option is turned off by default.

=item --icon

Path to an image file which will be included as part of the header in 
the top left of each page.

Image files may be in any format supported by L<PDF::API2>.

=item --icon-scale

Scaling value for icon images, default is 0.25.

=item --header | --noheader | --notitle

Prints a header consististing of the page title, and optionally
the timestamp and an image icon at the top of each page.
This option is enabled by default, use C<--notitle> or 
C<--noheader> to disable.

=item --footer | --nofooter | --page-numbers | --nopage-numbers

Adds the current page number to the bottom of each page. This is enabled
by default, use C<--nofooter> or C<--nopage-numbers> to disable.

=item --line-numbers | --noline-numbers

By default, line numbers will be included in the output PDF file. To
disable this behaviour, use the C<--noline-numbers> option.

=item --perl-syntax | --noperl-syntax

Enables or disables (default is enabled) Perl syntax highlighting. This
feature requires that the Perl::Tidy module is installed.

=item --page-width

=item --page-height

Page width and height in points. Default page size is 595 x 842 (A4).

=item --margins

=item --left-margin

=item --right-margin

=item --top-margin

=item --bottom-margin

Specifies the non-printable area of the page. The C<--margin> option will set
all margins to the same value, however individual margins may be altered with
the appropriate options. Values must be given in points. The default value for
the left and right margins is 48 points, and for the top and bottom margins is
60 points.

=item --font-face

Sets the font to use for the PDF file - currently this must be one of the PDF
core fonts. The default font face is Courier.

=item --font-size

Font size in points, default value is 10.

=item --line-spacing

Line spacing in points, default value is the font size + 2.

=item --noformfeed

By default, any formfeed characters in the input stream will be processed and
will act as expected, i.e. a new page will be started in the output PDF file.
This can be disabled with the C<--noformfeed> option which will cause all
formfeed characters to be ignored.

=back

Options may be given in any format recognised by the I<Getopt::Long> Perl
module, e.g. C<--name=value> or C<--name value>. Option names may be
abbreviated to their shortest unique value.

If the input filename is not given, then B<a2pdf> will expect to
receive input from STDIN.

=head1 DEPENDENCIES

B<a2pdf> requires the I<PDF::API2> Perl module (tested with PDF::API2
version 0.60).

Perl syntax highlighting requires the I<Perl::Tidy> module (tested with
Perl::Tidy version 20070504).

To include images in the page header, the modules I<File::Type> and
I<Image::Size> must be installed.

=head1 BUGS / ISSUES

=over 4

=item *

If the Perl syntax highlighting feature is used and the input Perl code
uses source filter modules, then depending on the changes made by the
source filter the syntax highlighting may not be performed correctly. 

=item *

When running under Red Hat 9, the LANG environment variable must be set
to 'C'.

=back

=head1 SEE ALSO

B<a2pdf> homepage - L<http://perl.jonallen.info/projects/a2pdf>

PDF::API2 - L<http://search.cpan.org/dist/PDF-API2>

Perl::Tidy - L<http://search.cpan.org/dist/Perl-Tidy>

File::Type - L<http://search.cpan.org/dist/File-Type>

Image::Size - L<http://search.cpan.org/dist/Image-Size>

=head1 AUTHOR

Written by Jon Allen (JJ), <jj@jonallen.info> / L<http://perl.jonallen.info>

=head1 COPYRIGHT and LICENCE

Copyright (C) 2007 Jon Allen (JJ) <jj@jonallen.info>

This software is licensed under the terms of the Artistic
License version 2.0.

For full license details, please read the file 'artistic-2_0.txt' 
included with this distribution, or see
http://www.perlfoundation.org/legal/licenses/artistic-2_0.html

=cut








a2pdf-1.13/Changes

Changelog for a2pdf (see http://perl.jonallen.info/projects/a2pdf)

1.13 - 07 April 2007
  Fixed minor formatting errors

1.12 - 06 April 2007
  Fixed inconsistent version numbering

1.09 - 06 April 2007
  Updated for compatibility with latest version of PDF::API2 (0.60)
  Changed license to Artistic 2.0
  Split into script (a2pdf) and module (App::a2pdf)

1.08 - 17 September 2004
  Now compatible with Windows as well as UNIX
  Added support for formfeed characters in input stream
    (thanks to Trevor Richardson)
  Page title can be specified (--title option)
  Image icon can be added to page header (--icon option)
  Added '--timestamp' option

1.07 - 14 April 2004
  Added '--help' and '--doc' options
  Documentation update
  Initial release to CPAN

1.06 - 03 April 2004
  Added '--version' option
  Changed header to include the modification timestamp of the 
  input file
  Words which are longer than the width of a complete line are
  now split instead of being printed in a smaller font size

1.05 - 28 March 2004
  Fixed bug which prevented header/footer display when the
  font size was >10

1.04 - 28 March 2004
  Added font cache to reduce the size of created PDF files
  Renamed script from a2pdf.pl to a2pdf
  Distribution created with the 'scriptdist' utility
  Documentation update

1.03 - 21 March 2004
  Fixed layout bug which occurred when the --noline-numbers
  option was used

1.02 - 20 March 2004
  Implemented Perl syntax highlighting (using Perl::Tidy)
  Both '-' and '_' characters are now allowed in option names
  (e.g. --perl-syntax is equivalent to --perl_syntax)

1.01 - 07 March 2004
  Added display of page numbers
  Increased default margin to 48 points

1.00 - 06 March 2004
  First public release

# App::a2pdf
#
# Copyright (C) 2007 Jon Allen <jj@jonallen.info>
#
# This software is licensed under the terms of the Artistic
# License version 2.0.
#
# For full license details, please read the file 'artistic-2_0.txt' 
# included with this distribution, or see
# http://www.perlfoundation.org/legal/licenses/artistic-2_0.html

package App::a2pdf;

use strict;
use warnings;
use PDF::API2;
use Switch 'Perl6';

BEGIN {
}


#-----------------------------------------------------------------------

sub new {
  my $invocant = shift;
  my $class    = ref($invocant) || $invocant;;

  # Set default options
  my $self = { @_ }; 
  bless $self,$class;
  
  # Define style mapping
  # This will relate Perl::Tidy's token types to a printing style
  $self->{stylemap} = {
    'header'     => 'helvetica_bold_10',
    'footer'     => 'helvetica_bold_10',
    'k'          => 'black_bold',
    '{'          => 'black_bold',
    '}'          => 'black_bold',
    'POD'        => 'grey_italic',
    'POD_START'  => 'grey_italic',
    'POD_END'    => 'grey_italic',
    'END_START'  => 'grey_italic',
    'DATA_START' => 'grey_italic',
    'DATA'       => 'grey_italic',
    'SYSTEM'     => 'grey_italic',
    '#'          => 'grey_italic',
    'J'          => 'red_italic',
    'j'          => 'red_italic',
    'i'          => 'blue',
    '->'         => 'blue',
    'w'          => 'green',
    'L'          => 'brown',
    'R'          => 'brown',
    'Q'          => 'purple',
    'q'          => 'purple',
  };
  
  # Define styles
  # Supports 3 properties, font (e.g. Helvetica, Courier, Times),
  # color (in hex), and type (Bold, Oblique, or BoldOblique)
  $self->{stylist} = {
    'helvetica9'  => {font=>'Helvetica',size=>9},
    'helvetica_bold_10'  => {font=>'Helvetica',size=>10,type=>'Bold'},
    'black_bold'  => {color=>'#000000',type=>'Bold'},
    'grey_italic' => {color=>'#333333',type=>'Oblique'},
    'red_italic'  => {color=>'#cc2222',type=>'Oblique'},
    'blue'        => {color=>'#222288'},
    'green'       => {color=>'#228822'},
    'brown'       => {color=>'#666622'},
    'purple'      => {color=>'#882288'},
  };
  
  # Set up first page
  $self->{page_number}   = 0;
  $self->{line_number}   = 1;
  $self->{line_number_width} = 0;
  $self->{line_spacing}  = $self->{font_size}+2 unless ($self->{line_spacing});
  $self->{x_position}    = $self->{left_margin};
  $self->{y_position}    = $self->{page_height} - $self->{top_margin};
  $self->{pdf}           = PDF::API2->new;
  $self->{pdf}->mediabox($self->{page_width},$self->{page_height});
  
  if ($self->{icon}) {
    # Load required modules to handle images
    eval "use File::Type;use Image::Size";
    unless ($@) {
      if (-e $self->{icon}) {
        my $type = File::Type->new->checktype_filename($self->{icon});
        given ($type) {
          when 'image/jpeg'  {$self->{icon_img} = $self->{pdf}->image_jpeg($self->{icon})}
          when 'image/tiff'  {$self->{icon_img} = $self->{pdf}->image_tiff($self->{icon})}
          when 'image/gif'   {$self->{icon_img} = $self->{pdf}->image_gif($self->{icon})}
          when 'image/x-png' {$self->{icon_img} = $self->{pdf}->image_png($self->{icon})}
          when 'image/x-pnm' {$self->{icon_img} = $self->{pdf}->image_pnm($self->{icon})}
          default {warn "[Warning] Unknown image format '$type' for icon ".$self->{icon}."\n"}
        }
        if ($self->{icon_img}) {
          ($self->{icon_width},$self->{icon_height}) = imgsize($self->{icon});  
        }    
      } else {
        warn("[Warning] Cannot open icon file: ".$self->{icon}."\n");
      }
    } else {
      warn("[Warning] The modules File::Type and Image::Size are required to use icons\n")
    }
  }

  $self->formfeed;
  $self->set_style;
  return $self;
}


#-----------------------------------------------------------------------

sub print {
  my $self = shift;
  my $line = shift;

  if ($self->{newline_flag}) {
    $self->newline;
    $self->{newline_flag} = 0;
  }

  $self->print_line_number if $self->{line_numbers};
  $self->print_text_with_style($line);
  $self->{newline_flag} = 1;
}


#-----------------------------------------------------------------------



sub write_line {    # This is the write_line method called by Perl::Tidy
  my $self        = shift;
  my $line        = shift;
  my $line_number = $line->{_line_number};
  my $line_type   = $line->{_line_type};
  my $line_text   = $line->{_line_text};
  chomp $line_text;

  if ($self->{newline_flag}) {
    $self->newline;
    $self->{newline_flag} = 0;
  }
  $self->print_line_number if $self->{line_numbers};

  if ($line_type eq 'CODE') {
    $self->print_text_with_style($1) if ($line_text =~ /^(\s+)/);
    my @rtoken_list  = @{$line->{_rtokens}};
    my @rtoken_types = @{$line->{_rtoken_type}};
    foreach my $rtoken (@rtoken_list) {
      my $rtoken_type = shift @rtoken_types;
      $self->print_text_with_style($rtoken,$rtoken_type);
    }
  } else {
    $self->print_text_with_style($line_text,$line_type);
  }
  $self->{newline_flag} = 1;
}


#-----------------------------------------------------------------------

sub newline {
  my $self = shift;
  
  $self->linefeed;
  $self->{line_number}++;
  $self->{overflow} = 0;
}


#-----------------------------------------------------------------------

sub linefeed {
  my $self = shift;
  
  $self->{y_position} -= $self->{line_spacing};
  $self->{x_position}  = $self->{left_margin} + $self->{line_number_width};
  $self->{overflow}    = 1;
    
  if ($self->{y_position} < ($self->{bottom_margin} + $self->{footer_height})) {
    my $style = $self->{style};
    $self->formfeed;
    $self->set_style($style);
  } 
}


#-----------------------------------------------------------------------

sub formfeed {
  my $self = shift;
  $self->{page}       = $self->{pdf}->page;
  $self->{x_position} = $self->{left_margin} + $self->{line_number_width};
  $self->{page_number}++;

  delete $self->{text};
  delete $self->{gfx};  
  $self->{gfx}  = $self->{page}->gfx;
  $self->{text} = $self->{page}->text;
  
  $self->{y_position}    = $self->{page_height} - $self->{top_margin} - $self->{line_spacing};
  $self->{header_height} = ($self->{header}) ? $self->generate_header : 0;
  $self->{footer_height} = ($self->{footer}) ? $self->generate_footer : 0;
  $self->{y_position}   -= $self->{header_height};
}


#-----------------------------------------------------------------------

sub generate_header {
  my $self = shift;
  $self->set_style('header');
  
  my $header_padding = 2;
  my $header_spacing = 3;
  my $header_height  = $self->{text_size} + $header_spacing + $header_padding;
  
  # Draw header icon
  if ($self->{icon_img}) {
    my $icon_height_in_points = $self->{icon_height} * $self->{icon_scale};
    if ($icon_height_in_points > $self->{text_size}) {
      $header_height += ($icon_height_in_points - $self->{text_size});
    }
    my $ypos = $self->{page_height} - $self->{top_margin} - $icon_height_in_points;
    $self->{gfx}->image($self->{icon_img},$self->{left_margin},$ypos,$self->{icon_scale});
  }

  # Add page title
  my $x = $self->{page_width} - $self->{right_margin} - $self->{text}->advancewidth($self->{title});
  my $y = $self->{page_height} - $self->{top_margin} - $header_height + $header_spacing + $header_padding;
  $self->{text}->textlabel($x,$y,$self->{fontcache}->{$self->{font}},$self->{text_size},$self->{title});

  # Draw horizontal line
  $self->{gfx}->move($self->{left_margin},$self->{page_height}-$self->{top_margin}-$header_height + $header_padding);
  $self->{gfx}->line($self->{page_width}-$self->{right_margin},$self->{page_height}-$self->{top_margin}-$header_height + $header_padding);
  $self->{gfx}->stroke;
  
  return $header_height;
}


#-----------------------------------------------------------------------

sub generate_footer {
  my $self  = shift;
  $self->set_style('footer');

  # Add page footer
  my $t = 'Page '.$self->{page_number};
  my $x = $self->{page_width} - $self->{right_margin} - $self->{text}->advancewidth($t);
  my $y = $self->{bottom_margin};
  $self->{text}->textlabel($x,$y,$self->{fontcache}->{$self->{font}},$self->{text_size},$t);


  $self->{gfx} = $self->{page}->gfx unless (exists $self->{gfx});
  $self->{gfx}->move($self->{left_margin},$self->{bottom_margin}+10);
  $self->{gfx}->line($self->{page_width}-$self->{right_margin},$self->{bottom_margin}+10);
  $self->{gfx}->stroke;
  return 18;  # Footer height in points
}


#-----------------------------------------------------------------------

sub output {
  my $self = shift;
  print $self->{pdf}->stringify;
  #$self->{pdf}->end;
}


#-----------------------------------------------------------------------

sub line_number_chars {
  my $self                      = shift;
  my $line_number_chars         = shift;
  $self->{line_number_chars}    = $line_number_chars;
  $self->{line_number_width}    = ($self->{line_numbers}) ? $self->{text}->advancewidth('X' x ($line_number_chars + 2)) : 0;
  $self->{line_number_template} = '%'.$line_number_chars.'d: %s';
  $self->{x_position}           = $self->{left_margin} + $self->{line_number_width};
}


#-----------------------------------------------------------------------

sub print_line_number {
  my $self = shift;
  
  $self->set_style;
  my $width = $self->{text}->advancewidth($self->{line_number}.':X');
  my $x_pos = $self->{left_margin} + $self->{line_number_width} - $width;
  $self->{text}->textlabel($x_pos,$self->{y_position},$self->{fontcache}->{$self->{font}},$self->{text_size},$self->{line_number}.':');
}


#--print_text_with_style---------------------------------------------------

sub print_text_with_style {
  my $self  = shift;
  my $text  = shift;
  $self->set_style(shift);

  while ($text =~ /(\f|[^\f]+)/g) {
    my $block = $1;
    if ($block =~ /\f/ && !exists $self->{noformfeed}) {
      $self->formfeed;
      $self->{x_position} = $self->{left_margin} + $self->{line_number_width};
    } else {
      while ($block =~ /(\s+|\S+)/g) {
        my $word = $1;
        $self->print_word($word);
      }
    }
  }
}


#--print_word--------------------------------------------------------------
#
# Purpose: Adds a single word to the PDF in the current style
#
# Usage:   $self->print_word('word');
#
#--------------------------------------------------------------------------

sub print_word {
  my $self = shift;
  my $word = shift;
  

    my $width = $self->{text}->advancewidth($word);
    if ($self->{x_position} + $width > $self->{page_width} - $self->{right_margin}) {
      # If the word will not fit on one line, split it up and recurse the 'print_word' sub
      if ($width > ($self->{page_width} - $self->{left_margin} - $self->{right_margin})) {
        my $fit = int(($self->{page_width} - $self->{x_position} - $self->{right_margin}) / $self->{nspace});
        my @words = (substr($word,0,$fit),substr($word,$fit));
        $self->print_word($_) foreach @words; 
        return;
      }
      $self->linefeed;
      if ($word =~ /^\s+$/ &&
          $self->{overflow} &&
          $self->{x_position} == $self->{left_margin} + $self->{line_number_width}) {
        return;
      }
    }
    $self->{x_position} += $self->{text}->textlabel($self->{x_position},
                                                    $self->{y_position},
                                                    $self->{fontcache}->{$self->{font}},
                                                    $self->{text_size},
                                                    $word,
                                                    -color => $self->{text_color});
    if ($self->{x_position} > $self->{page_width} - $self->{right_margin}) {
      $self->linefeed;
    }
}


#--set_style---------------------------------------------------------------
#
# Purpose: Sets current style (font, size, colour)
#
# Usage:   $self->set_style('stylename');
#
#--------------------------------------------------------------------------

sub set_style {
  my $self  = shift;
  my $style = shift || 'default';
  
  $style = (exists $self->{stylemap}->{$style}) ? $self->{stylemap}->{$style} : 'default';
  
  # Create font object if necessary
  my $font = ($self->{stylist}->{$style}->{font} || $self->{font_face}) .
             ((exists $self->{stylist}->{$style}->{type}) ? '-'.$self->{stylist}->{$style}->{type} : '');
  unless (exists $self->{fontcache}->{$font}) {
    $self->{fontcache}->{$font} = $self->{pdf}->corefont($font);
  }
  
  $self->{style}      = $style;
  $self->{font}       = $font;
  $self->{text_color} = $self->{stylist}->{$style}->{color} || '#000000';
  $self->{text_size}  = $self->{stylist}->{$style}->{size}  || $self->{font_size};
  
  $self->{text}->font($self->{fontcache}->{$font},$self->{text_size});
  $self->{nspace}     = $self->{text}->advancewidth('n');
}


#--------------------------------------------------------------------------

sub _MANIFEST {
  require File::Type;
  require Image::Size;
  require PDF::API2::Content;
  require PDF::API2::Win32;
  require PDF::API2::Lite;
  require PDF::API2::UniWrap;
}

1;







a2pdf-1.13/Makefile.PL

# $Id$
use ExtUtils::MakeMaker;

eval "use Test::Manifest";

unless( $@ )
	{
	no warnings;
	
	*ExtUtils::MM_Any::test_via_harness = sub
		{
		my($self, $perl, $tests) = @_;
	
		return qq|	$perl "-MTest::Manifest" | .
			qq|"-e" "run_t_manifest(\$(TEST_VERBOSE), '\$(INST_LIB)', | .
			qq|'\$(INST_ARCHLIB)')"\n|;
		};
	}



WriteMakefile(
		'NAME'         => 'a2pdf',
                'AUTHOR'       => 'Jon Allen <jj@jonallen.info>',
                'ABSTRACT'     => 'Converts ASCII text to PDF format, with optional line/page numbering and Perl syntax highlighting',
                'VERSION_FROM' => 'bin/a2pdf',
		'EXE_FILES'    => [ 'bin/a2pdf' ],            
		'PREREQ_PM'    => { 'PDF::API2' => 0.60, },
		'MAN1PODS'     => {
			'bin/a2pdf' => "\$(INST_MAN1DIR)/a2pdf.1",
			},

		clean => { FILES => "*.bak a2pdf-*" },
		);
        
1;







a2pdf-1.13/MANIFEST

Changes
MANIFEST			This list of files
README
Makefile.PL
artistic-2_0.txt
bin/a2pdf
lib/App/a2pdf.pm
t/compile.t
t/pod.t
t/test_manifest
META.yml                                 Module meta-data (added by MakeMaker)







a2pdf-1.13/META.yml

# http://module-build.sourceforge.net/META-spec.html
#XXXXXXX This is a prototype!!!  It will change in the future!!! XXXXX#
name:         a2pdf
version:      1.13
version_from: bin/a2pdf
installdirs:  site
requires:
    PDF::API2:                     0.6

distribution_type: module
generated_by: ExtUtils::MakeMaker version 6.30







a2pdf-1.13/README

NAME
  a2pdf - converts ASCII text to PDF format, with optional
  page/line numbering and Perl syntax highlighting

INSTALLATION
  Unzip and untar the archive, cd to the distribution
  directory and enter the following commands:

    perl Makefile.PL
    make
    make test
    make install

USAGE
  a2pdf [options] input.txt >output.pdf
  Options are described in the main a2pdf documentation

DEPENDENCIES
  Requires the PDF::API2 module, and optionally Perl::Tidy
  to perform Perl syntax highlighting

AUTHOR
  Written by Jon Allen (JJ) <jj@jonallen.info>
  http://perl.jonallen.info/projects/a2pdf








a2pdf-1.13/t/compile.t

# $Id$

use Test::More tests => 1;

my $file = "blib/script/a2pdf";

print "bail out! Script file is missing!" unless -e $file;

my $output = `perl -c $file 2>&1`;

print "bail out! Script file is missing!" unless
	like( $output, qr/syntax OK$/, 'script compiles' );







a2pdf-1.13/t/pod.t

# $Id$
use Test::More;
eval "use Test::Pod 1.00";
plan skip_all => "Test::Pod 1.00 required for testing POD" if $@;
all_pod_files_ok();







a2pdf-1.13/t/test_manifest

compile.t
pod.t






